




Faculdade de Ciências Exatas e da Engenharia

2019/2020

Arquitetura de Computadores

Licenciatura em Engenharia Informática

1º Projeto – Processador Básico



Trabalho realizado por:

Diego Briceño (nº 2043818)

Rúben Rodrigues (nº 2046018)

Funchal, 14 de março de 2020

Índice

1.	Introdução	3
2.	Objetivos.....	3
3.	Desenvolvimento	3
3.1.	Placa-mãe.....	3
3.1.1.	Memória de Dados (RAM)	3
3.1.2.	Memória de Instruções.....	3
3.1.3.	Processador.....	4
3.1.3.1.	Periférico de Entrada.....	4
3.1.3.2.	Periférico de Saída	4
3.1.3.3.	Multiplexer dos Registos (Mux R).....	4
3.1.3.4.	Registos A e B	4
3.1.3.5.	Unidade Aritmética e Lógica (ALU).....	5
3.1.3.6.	Comparação	5
3.1.3.7.	Contador de programa (PC).....	6
3.1.3.8.	<i>Multiplexer do Program Counter</i> (Mux_PC)	6
3.1.3.9.	ROM de descodificação (ROM)	6
4.	Discussão de Resultados.....	7
5.	Conclusão	7
6.	Bibliografia.....	7
7.	Anexo A	8
7.1.	Tabela de Instruções de teste.....	8
7.2.	Fluxograma	9
7.3.	Simulação com o $PIN < 0$	10
7.4.	Simulação com o $PIN \geq 40$	10
7.5.	Simulação com o $0 \leq PIN < 40$	11
8.	Anexo B.....	12
8.1.	Placa-Mãe	12
8.1.1.	Processador.....	12
8.1.1.1.	Periférico de Entrada.....	14
8.1.1.2.	Periférico de Saída	14
8.1.1.3.	Multiplexer dos Registos (Mux R).....	14
8.1.1.4.	Registos A e B	15
8.1.1.5.	Unidade Aritmética e Lógica (ALU).....	15
8.1.1.6.	Comparação	16
8.1.1.7.	Contador de programa (PC).....	16
8.1.1.8.	Multiplexer do Program Counter (Mux_PC).....	17
8.1.1.9.	ROM de descodificação (ROM)	17
8.1.2.	Memória de Instruções.....	18
8.1.3.	Memória de Dados (RAM)	19

1. Introdução

Este relatório apresentará os objetivos relacionados ao primeiro trabalho prático da unidade curricular de Arquitetura de Computadores assim como o seu desenvolvimento, discussão de resultados e a conclusão a que os alunos chegaram no fim.

Os processadores são as unidades centrais dos sistemas computacionais. Se comparássemos um sistema computacional a uma pessoa, o processador seria o “cérebro” pois é ele que executa as instruções de máquina, que são qualquer tarefa que o processador possa executar utilizando uma série de cálculos e decisões.

2. Objetivos

O objetivo deste trabalho é realizar um processador básico, com um conjunto mínimo de instruções, em linguagem de descrição de hardware (VHDL). Para este fim, utilizou-se o programa ISE da Xilinx com a simulação sendo efetuada no ISim e o teste em FPGA (Spartan 3E e Artix 7).

3. Desenvolvimento

O processador desenvolvido é constituído por vários módulos que, quando conectados com a memória de dados e a memória de instruções, formam a placa mãe. Cada módulo foi implementado em separado, como será descrito de seguida, de modo a facilitar a implementação, sendo só necessário no fim ligar os diferentes módulos.

3.1. Placa-mãe

3.1.1. Memória de Dados (RAM)

A RAM (Random-Access Memory) é um tipo de memória que permite a escrita/leitura de dados, individual e aleatória (como o nome indica), através do seu endereço. É uma memória volátil, ou seja, os seus dados perdem-se quando a memória perde a alimentação elétrica.

No caso desta placa-mãe, a memória de dados guarda os dados presentes no sinal de entrada **Operando1**, de 8 bits, quando o sinal **WR** está a ‘1’ na transição ascendente do sinal de relógio (**clk**), no endereço indicado pelo sinal de entrada **Constante**, de 8 bits. Quando o sinal **WR** está a ‘0’ é feita a leitura dos dados, na posição de memória indicada por **Constante** e o valor lido é atribuído ao sinal de saída **Dados_M**, de 8 bits.

3.1.2. Memória de Instruções

É neste módulo que ficam armazenadas as instruções do programa a ser executado. Apresenta uma dimensão de 14 bits, onde o endereço da instrução é determinado pelo sinal **Endereço**, de 8 bits, e à saída é disponibilizado o **opcode**, de 5 bits, o sinal **SEL_R**, de 1 bit, e o sinal **Constante**, de 8 bits.

3.1.3.Processador

3.1.3.1. Periférico de Entrada

É neste módulo que é feita a comunicação do processador com o exterior, permitindo ao utilizador inserir dados para posteriormente serem realizadas operações com os mesmos. Alguns exemplos deste tipo de periféricos são o teclado e o rato.

Este módulo é controlado pelo sinal $\overline{ESCR_P}$, de 1 bit, que quando está a '1' é feita uma leitura dos dados de entrada, PIN , de 8 bits, colocando-os na saída do periférico, $Dados_IN$, de 8 bits.

3.1.3.2. Periférico de Saída

Este módulo permite que o utilizador veja os dados e informações processados pelo computador. Alguns exemplos deste tipo de periféricos são o monitor, a impressora e colunas de som. Este periférico é controlado pelo sinal $ESCR_P$, de 1 bit, que quando está a '1', na transição ascendente do relógio (clk), escreve no sinal de saída, $POUT$, de 8 bits, o valor do sinal à entrada do módulo, $Operando1$, também de 8 bits.

3.1.3.3. Multiplexer dos Registos (Mux R)

Este módulo é responsável por encaminhar um dos quatro sinais disponíveis, de 8 bits, à sua entrada ($Resultado$, $Dados_IN$, $Dados_M$ e $Constante$) para apresentar na sua saída, $Dados_R$, de 8 bits. O sinal a encaminhar depende do valor do sinal de entrada SEL_Data , de 2 bits.

SEL_DATA	DADOS_R
00	Resultado
01	Dados_IN
10	Dados_M
11	Constante

Tabela 1 Sinal de saída do Mux R em função do sinal SEL_DATA.

3.1.3.4. Registos A e B

A escrita nos registos A e B é controlada pelo sinal $ESCR_R$, de 1 bit. Quando o sinal está a '1' o valor presente no sinal de entrada $Dados_R$, de 8 bits, é guardado no registo especificado pelo sinal SEL_R , de 1 bit, na transição ascendente do sinal de relógio (clk). Estes registos estão continuamente a efetuar leituras. As saídas $Operando1$ e $Operando2$, ambas de 8 bits, apresentam os valores guardados nos registos A e B, respetivamente.

SEL_R	Registo a ser escrito
0	Registo A
1	Registo B

Tabela 2 Sinal de seleção de escrita nos registos A e B

3.1.3.5. Unidade Aritmética e Lógica (ALU)

Este módulo permite realizar operações aritméticas e lógicas, tal como o nome indica. No caso desta placa-mãe, a unidade aritmética e lógica do processador, é capaz de realizar as operações soma, subtração, AND, OR e XOR, com os sinais de entrada **Operando1** e **Operando2**, ambos de 8 bits, que representam números inteiros com sinal. Os sinais de saída da ALU são determinados pelo sinal de seleção **SEL_ALU**, de 3 bits, A saída **Resultado**, de 8 bits, será atualizada no caso de cada operação e, a saída **COMP_RES**, de 5 bits, será atualizada apenas quando é realizada uma comparação, cada um dos seus bits indicando o resultado de uma das cinco comparações apresentadas no módulo *Comparação*.

SEL_ALU	Operação
000	Operando1 + Operando2
001	Operando1 – Operando2
010	Operando1 AND Operando2
011	Operando1 OR Operando2
100	Operando1 XOR Operando3
101	Operando1 > Operando2 Operando1 >= Operando2 Operando1 = Operando2 Operando1 <= Operando2 Operando1 < Operando2

Tabela 3 Operações da ALU

3.1.3.6. Comparação

O funcionamento deste módulo é semelhante ao módulo dos registos. Guarda o sinal de entrada, que neste caso é o sinal **COMP_RES**, sinal de 5 bits, quando o sinal **COMP_FLAG** está a ‘1’ e o sinal de relógio encontra-se na transição ascendente. Este módulo está constantemente a efetuar leituras, mas apenas um dos 5 bits do sinal guardado é encaminhado para a saída, **S_FLAG**, de 1 bit. O sinal de seleção **SEL_COMP**, de 3 bits, determina qual o bit guardado que está disponível na saída, do modo apresentado na tabela abaixo.

SEL_COMP	S_FLAG
000	COMP_RES(0) (>)
001	COMP_RES(1) (>=)
010	COMP_RES(2) (=)
011	COMP_RES(3) (<=)
100	COMP_RES(4) (<)

Tabela 4 Sinal de saída do multiplexador de comparação em função do sinal de seleção SEL_COMP

3.1.3.7. Contador de programa (PC)

O contador de programa indica qual é a posição atual da sequência de execução de um programa. Na transição ascendente do relógio, a saída **Endereço**, de 8 bits, é enviada à Memória de Instruções. A sequência de execução será incrementada de um em um quando a entrada **ESCR_PC**, de 1 bit, estiver a '0', caso contrário, a saída do contador corresponderá ao valor da entrada **Constante**, de 8 bits, e neste caso ocorrerá um salto para o endereço de instrução indicado por este sinal. A entrada **Reset**, de 1 bit, permite voltar ao início do programa quando ativa.

3.1.3.8. Multiplexer do Program Counter (Mux_PC)

Este *multiplexer* indica ao contador de programa se é para realizar um salto ou simplesmente incrementar o contador, através do sinal de saída **ESCR_PC**, de 1 bit, como já foi visto no módulo do Contador de programa.

O sinal de seleção deste *multiplexer* é o sinal **SEL_PC**, de 3 bits, que indica qual dos valores de entrada deve passar para a saída, como indicado na tabela seguinte.

SEL_PC	ESCR_PC
000	'0'
001	'1'
010	S_FLAG
011	Operando1(7)
100	NOT (Operando1(7) OR Operando1(6) OR Operando1(5) OR Operando1(4) OR Operando1(3) OR Operando1(2) OR Operando1(1) OR Operando1(0))

Tabela 5 Valor de saída do MUX_PC em função do sinal de seleção SEL_PC

3.1.3.9. ROM de decodificação (ROM)

Esta ROM é responsável por fornecer aos restantes módulos os seus sinais de controlo. Esta recebe o sinal **opcode**, de 5 bits, da memória de instruções e coloca na sua saída os valores correspondentes aos seguintes sinais de controlo: **SEL_PC**, **SEL_COMP** e **SEL_ALU**, de 3 bits, **SEL_Data**, de 2 bits, e os sinais **COMP_FLAG**, **ESCR_R**, **ESCR_P** e **WR**, de 1 bit. Na tabela 6 presente no enunciado encontra-se a relação entre o sinal **opcode** e os sinais de controlo, onde cada instrução está também indicada em linguagem *assembly*. De considerar que o *Ri* corresponde ao registo indicado pelo sinal **SEL_R**, de 1 bit.

No código é utilizado um *case* para implementar a tabela referida anteriormente.

4. Discussão de Resultados

Para saber o procedimento que o processador efetuará quando são executadas as instruções mostradas do teste no enunciado, cuja tabela encontra-se no Anexo A, foi preciso traduzir as instruções que se encontravam em linguagem *assembly* para código máquina de modo a programar a memória de instruções.

O teste referido está feito de maneira a que o processador siga as seguintes operações:

- Quando o PIN a ser introduzido representa um valor negativo, o programa realiza a operação:

$$\text{POUT} = - \text{PIN}$$

- ❖ Esta situação encontra-se representada na simulação presente no Anexo A, ponto 7.3.

- Quando o PIN a ser introduzido representa um valor maior ou igual a 40, o programa realiza a operação:

$$\text{POUT} = \text{PIN} - 20$$

- ❖ Esta situação encontra-se representada na simulação presente no Anexo A, ponto 7.4.

- Caso contrário, o PIN é um valor positivo menor que 40, o processador realiza a operação:

$$\text{POUT} = 3 * \text{PIN}$$

- ❖ Esta situação encontra-se representada na simulação presente no Anexo A, ponto 7.5.

O programa encontra-se esquematizado através de um fluxograma no ponto 7.2 do Anexo A.

5. Conclusão

Para concluir, este projeto permitiu aos alunos ganhar uma melhor compreensão sobre o funcionamento interno do computador, em particular o processador. As simulações no ISim e os testes nas FPGA resultaram como era esperado, confirmando-se assim o sucesso durante a elaboração deste primeiro projeto.

6. Bibliografia

J. Delgado e C. Ribeiro, Arquitectura de Computadores, FCA - Editora de Informática, 2010.

7. Anexo A

7.1. Tabela de Instruções de teste

	Endereço	Instrução (Assembly)	Instrução (código máquina)		
			Opcode	SEL_R	Constante
0	00000000	LD RA, 3	00010	0	00000011
1	00000001	ST [0], RA	00100	0	00000000
2	00000010	LD RA, 20	00010	0	00010100
3	00000011	ST [1], RA	00100	0	00000001
4	00000100	LDP RA	00000	0	XXXXXXXXXX
5	00000101	JN 22	10001	X	00010110
6	00000110	LD RB, 40	00010	1	00101000
7	00000111	CMP RA, RB	01010	X	XXXXXXXXXX
8	00001000	JGE 27	01100	X	00011011
9	00001001	ST [2], RA	00100	0	00000010
10	00001010	ST [3], RA	00100	0	00000011
11	00001011	LD RA, [0]	00011	0	00000000
12	00001100	LD RB, 1	00010	1	00000001
13	00001101	SUB RA, RB	00110	0	XXXXXXXXXX
14	00001110	JZ 20	10010	X	00010100
15	00001111	ST [0], RA	00100	0	00000000
16	00010000	LD RA, [3]	00011	0	00000011
17	00010001	LD RB, [2]	00011	1	00000010
18	00010010	ADD RA, RB	00101	0	XXXXXXXXXX
19	00010011	JMP 10	10000	X	00001010
20	00010100	LD RA [3]	00011	0	00000011
21	00010101	JMP 29	10000	X	00011101
22	00010110	LD RB, -1	00010	1	11111111
23	00010111	XOR RA, RB	01001	0	XXXXXXXXXX
24	00011000	LD RB, 1	00010	1	00000001
25	00011001	ADD RA, RB	00101	0	XXXXXXXXXX
26	00011010	JMP 29	10000	X	00011101
27	00011011	LD RB, [1]	00011	1	00000001
28	00011100	SUB RA, RB	00110	0	XXXXXXXXXX
29	00011101	STP RA	00001	X	XXXXXXXXXX
30	00011110	JMP 30	10000	X	00011110

Tabela 6 Instruções de teste do projeto

7.2. Fluxograma




Figura 1 Fluxograma do projeto

7.3. Simulação com o $PIN < 0$




Ilustração 1 Simulação com o valor de PIN a -2

7.4. Simulação com o $PIN \geq 40$




Ilustração 2 Simulação com o valor de PIN a 126

7.5. Simulação com o $0 \leq \text{PIN} < 40$




Ilustração 3 Simulação com o valor de PIN a 15

8. Anexo B

8.1. Placa-Mãe

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity PlacaMae is
5      Port ( Pin : in  STD_LOGIC_VECTOR (7 downto 0);
6            Clk : in  STD_LOGIC;
7            Reset : in  STD_LOGIC;
8            Pout : out STD_LOGIC_VECTOR (7 downto 0));
9  end PlacaMae;
10
11 architecture Struct of PlacaMae is
12     Component Memoria_Instrucoes is
13         Port (Endereco : in  STD_LOGIC_VECTOR (7 downto 0);
14               Opcode : out STD_LOGIC_VECTOR (4 downto 0);
15               Sel_R : out STD_LOGIC;
16               Constante : out STD_LOGIC_VECTOR (7 downto 0));
17     end Component;
18
19     Component Processador is
20         Port ( Pin,Dados_M, Const : in  STD_LOGIC_VECTOR (7 downto 0);
21               Clk, Reset, Sel_R : in  STD_LOGIC;
22               Opcode : in  STD_LOGIC_VECTOR (4 downto 0);
23               Pout, Endereco, Opl,Const_Out, ResulALU, Op2 : out STD_LOGIC_VECTOR (7 downto 0);
24               ResComp : out STD_LOGIC_VECTOR (4 downto 0);
25               WR : out STD_LOGIC);
26     end Component;
27
28     Component RAM is
29         Port ( Operandol, Address : in  STD_LOGIC_VECTOR (7 downto 0);
30               WR, Clock : in  STD_LOGIC;
31               Dados_M : out STD_LOGIC_VECTOR (7 downto 0));
32     end Component;
33
34     signal Select_Reg,WriteRead: STD_LOGIC;
35     signal Const,NAOIMPORTA,Instrucao, Dados_Mem, Operandol, ResulALU, Op2: STD_LOGIC_VECTOR (7 downto 0);
36     signal OPCODE, ResComp : STD_LOGIC_VECTOR (4 downto 0);
37     begin
38         Proc: Processador Port Map(Pin, Dados_Mem, Const, Clk, Reset, Select_Reg, OPCODE, Pout, Instrucao, Operandol, NAOIMPORTA, ResulALU, Op2, ResComp, WriteRead);
39         Mem_Instrucs: Memoria_Instrucoes Port Map(Instrucao,OPCODE,Select_Reg,Const);
40         Mem_Dados: RAM Port Map(Operandol,Const,WriteRead,Clk,Dados_Mem);
41
42     end Struct;
```

8.1.1. Processador

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity Processador is
5      Port ( Pin,Dados_M, Const : in  STD_LOGIC_VECTOR (7 downto 0);
6            Clk, Reset, Sel_R : in  STD_LOGIC;
7            Opcode : in  STD_LOGIC_VECTOR (4 downto 0);
8            Pout, Endereco, Opl,Const_Out, ResulALU, Op2 : out STD_LOGIC_VECTOR (7 downto 0);
9            ResComp : out STD_LOGIC_VECTOR (4 downto 0);
10           WR : out STD_LOGIC);
11 end Processador;
12
13 architecture Struct of Processador is
14     Component Comparacao is
15         Port ( Comp_Res : in  STD_LOGIC_VECTOR (4 downto 0);
16               Comp_Flag : in  STD_LOGIC;
17               Clk : in  STD_LOGIC;
18               Sel_Comp : in  STD_LOGIC_VECTOR (2 downto 0);
19               S_Flag : out STD_LOGIC);
20     end Component;
21
22     Component Mux_PC is
23         Port ( S_FLAG : in  STD_LOGIC;
24               Operandol : in  STD_LOGIC_VECTOR (7 downto 0);
25               SEL_PC : in  STD_LOGIC_VECTOR (2 downto 0);
26               ESCR_PC : out STD_LOGIC);
27     end Component;
28
29     Component Mux_Registros is
30         Port ( Resultado : in  STD_LOGIC_VECTOR (7 downto 0);
31               Dados_IN : in  STD_LOGIC_VECTOR (7 downto 0);
32               Dados_M : in  STD_LOGIC_VECTOR (7 downto 0);
33               Constante : in  STD_LOGIC_VECTOR (7 downto 0);
34               SEL_Data : in  STD_LOGIC_VECTOR (1 downto 0);
35               Dados_Reg : out STD_LOGIC_VECTOR (7 downto 0));
36     end Component;
37
38     Component Periferico_Entrada is
39         Port ( ESCR_P : in  STD_LOGIC;
40               PIN : in  STD_LOGIC_VECTOR (7 downto 0);
41               Dados_In : out STD_LOGIC_VECTOR (7 downto 0));
42     end Component;
```

```

44 Component Periferico_Saida is
45     Port ( Escr_P : in STD_LOGIC;
46           CLK : in STD_LOGIC;
47           Operand1 : in STD_LOGIC_VECTOR (7 downto 0);
48           POut : out STD_LOGIC_VECTOR (7 downto 0));
49 end Component;
50
51 Component ProgramCounter is
52     Port ( Constante : in STD_LOGIC_VECTOR (7 downto 0);
53           ESCR_PC,Clock,Reset : in STD_LOGIC;
54           Endereco : out STD_LOGIC_VECTOR (7 downto 0));
55 end Component;
56
57 Component ROM_Descodificacao is
58     Port ( Opcode : in STD_LOGIC_VECTOR (4 downto 0);
59           Sel_ALU : out STD_LOGIC_VECTOR (2 downto 0);
60           Escr_Perif_Saida : out STD_LOGIC;
61           Sel_Data : out STD_LOGIC_VECTOR (1 downto 0);
62           Escr_Registo : out STD_LOGIC;
63           WR : out STD_LOGIC;
64           Sel_PC : out STD_LOGIC_VECTOR (2 downto 0);
65           Comp_Flag : out STD_LOGIC;
66           Sel_Comp : out STD_LOGIC_VECTOR (2 downto 0));
67 end Component;
68
69 Component RegistosAeB is
70     Port ( ESCR_R : in STD_LOGIC;
71           Dados_R : in STD_LOGIC_VECTOR (7 downto 0);
72           SEL_R : in STD_LOGIC;
73           clk : in STD_LOGIC;
74           Operand1 : out STD_LOGIC_VECTOR (7 downto 0);
75           Operando2 : out STD_LOGIC_VECTOR (7 downto 0));
76 end Component;
77
78 Component ALU is
79     Port ( Operand1 : in STD_LOGIC_VECTOR (7 downto 0);
80           Operando2 : in STD_LOGIC_VECTOR (7 downto 0);
81           Sel_ALU : in STD_LOGIC_VECTOR (2 downto 0);
82           Resultado : out STD_LOGIC_VECTOR (7 downto 0);
83           Comp_Res : out STD_LOGIC_VECTOR (4 downto 0));
84 end Component;
85
86 signal escrPerifs,escrPC, S_Flag, Compa_Flag,Select_Reg,Escrive_Reg: STD_LOGIC;
87 signal DadosPIN, Opel,Ope2, ResultadoALU,Dados_REG: STD_LOGIC_VECTOR (7 downto 0);
88 signal Selec_PC, Selec_Comp, Selec_ALU : STD_LOGIC_VECTOR (2 downto 0);
89 signal Sel_Data : STD_LOGIC_VECTOR (1 downto 0);
90 signal Resultado_Comparacao : STD_LOGIC_VECTOR (4 downto 0);
91 begin
92     PerifIn: Periferico_Entrada Port Map(escrPerifs,Pin,DadosPIN);
93     PerifOut: Periferico_Saida Port Map(escrPerifs,Clk,Opel,Pout);
94     Multiplexer_PC: Mux_PC Port Map(S_Flag,Opel,Selec_PC,escrPC);
95     PC: ProgramCounter Port Map(Const,escrPC,Clk, Reset, Endereco);
96     Comp: Comparacao Port Map(Resultado_Comparacao,Compa_Flag,Clk,Selec_Comp,S_Flag);
97     MUX_Reg: Mux_Registos Port Map(ResultadoALU,DadosPIN,Dados_M,Const,Sel_Data,Dados_REG);
98     ArithLU: ALU Port Map(Opel,Ope2,Selec_ALU,ResultadoALU,Resultado_Comparacao);
99     RegAeB: RegistosAeB Port Map(Escrive_Reg,Dados_REG,Sel_R,Clk,Opel,Ope2);
100    ROM_Decode: ROM_Descodificacao Port Map(Opcode,Selec_ALU,escrPerifs,Sel_Data,Escrive_Reg,WR,Selec_PC,Compa_Flag,Selec_Comp);
101    Op1 <= Opel;
102    Const_Out <= Const;
103    ResultadoALU <= ResultadoALU;
104    Op2 <= Ope2;
105    ResComp <= Resultado_Comparacao;
106
107 end Struct;

```

8.1.1.1. Periférico de Entrada

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3
4 entity Periferico_Entrada is
5     Port ( ESCR_P : in  STD_LOGIC;
6           PIN : in  STD_LOGIC_VECTOR (7 downto 0);
7           Dados_In : out  STD_LOGIC_VECTOR (7 downto 0));
8 end Periferico_Entrada;
9
10 architecture Behavioral of Periferico_Entrada is
11
12 begin
13     process (ESCR_P, PIN)
14     begin
15         if (ESCR_P = '0') then
16             Dados_In <= PIN;
17         end if;
18     end process;
19
20
21 end Behavioral;
```

8.1.1.2. Periférico de Saída

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3
4 entity Periferico_Saida is
5     Port ( Escr_P : in  STD_LOGIC;
6           CLK : in  STD_LOGIC;
7           Operandol : in  STD_LOGIC_VECTOR (7 downto 0);
8           POut : out  STD_LOGIC_VECTOR (7 downto 0));
9 end Periferico_Saida;
10
11 architecture Behavioral of Periferico_Saida is
12     signal temp : STD_LOGIC_VECTOR (7 downto 0);
13
14 begin
15     process (CLK)
16     begin
17         if rising_edge (CLK) then
18             if (Escr_P = '1') then
19                 temp <= Operandol;
20             end if;
21         end if;
22     end process;
23     POut <= temp;
24
25
26
27 end Behavioral;
```

8.1.1.3. Multiplexer dos Registos (Mux R)

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3
4 entity Mux_Registos is
5     Port ( Resultado : in  STD_LOGIC_VECTOR (7 downto 0);
6           Dados_IN : in  STD_LOGIC_VECTOR (7 downto 0);
7           Dados_M : in  STD_LOGIC_VECTOR (7 downto 0);
8           Constante : in  STD_LOGIC_VECTOR (7 downto 0);
9           SEL_Data : in  STD_LOGIC_VECTOR (1 downto 0);
10           Dados_Reg : out  STD_LOGIC_VECTOR (7 downto 0));
11 end Mux_Registos;
12
13 architecture Behavioral of Mux_Registos is
14     signal temp : STD_LOGIC_VECTOR (7 downto 0);
15
16 begin
17     process (SEL_Data, Resultado, Dados_IN, Dados_M, Constante)
18     begin
19         case SEL_Data is
20             when "00" => temp <= Resultado;
21             when "01" => temp <= Dados_IN;
22             when "10" => temp <= Dados_M;
23             when "11" => temp <= Constante;
24             when others => temp <= (others => 'X');
25         end case;
26     end process;
27     Dados_Reg <= temp;
28
29
30 end Behavioral;
```


8.1.1.4. Registos A e B

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity RegistosAeB is
5      Port ( ESCR_R : in  STD_LOGIC;
6            Dados_R : in  STD_LOGIC_VECTOR (7 downto 0);
7            SEL_R : in  STD_LOGIC;
8            clk : in  STD_LOGIC;
9            Operando1 : out STD_LOGIC_VECTOR (7 downto 0);
10           Operando2 : out STD_LOGIC_VECTOR (7 downto 0));
11 end RegistosAeB;
12
13 architecture Behavioral of RegistosAeB is
14
15 begin
16     process(ESCR_R, Dados_R, SEL_R, clk)
17     begin
18         if rising_edge(clk) then
19             if (ESCR_R = '1') then
20                 if (SEL_R = '0') then
21                     Operando1 <= Dados_R;
22                 else
23                     Operando2 <= Dados_R;
24                 end if;
25             end if;
26         end if;
27     end process;
28
29 end Behavioral;
```

8.1.1.5. Unidade Aritmética e Lógica (ALU)

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3  use IEEE.STD_LOGIC_ARITH.ALL;
4  use IEEE.STD_LOGIC_SIGNED.ALL;
5
6  entity ALU is
7      Port ( Operando1 : in  STD_LOGIC_VECTOR (7 downto 0);
8            Operando2 : in  STD_LOGIC_VECTOR (7 downto 0);
9            Sel_ALU : in  STD_LOGIC_VECTOR (2 downto 0);
10           Resultado : out STD_LOGIC_VECTOR (7 downto 0);
11           Comp_Res : out STD_LOGIC_VECTOR (4 downto 0));
12 end ALU;
13
14 architecture Behavioral of ALU is
15 begin
16     process(Operando1, Operando2, Sel_ALU)
17     begin
18         case Sel_ALU is
19             when "000" => Resultado <= Operando1 + Operando2;
20             when "001" => Resultado <= Operando1 - Operando2;
21             when "010" => Resultado <= Operando1 and Operando2;
22             when "011" => Resultado <= Operando1 or Operando2;
23             when "100" => Resultado <= Operando1 xor Operando2;
24             when "101" =>
25                 if (Operando1 > Operando2)
26                     then
27                         Comp_Res <= (0 => '1', 1 => '1', others => '0');
28                     end if;
29
30                 if (Operando1 = Operando2)
31                     then
32                         Comp_Res <= (1 => '1', 2 => '1', 3 => '1', others => '0');
33                     end if;
34
35                 if (Operando1 < Operando2)
36                     then
37                         Comp_Res <= (4 => '1', 3 => '1', others => '0');
38                     end if;
39             when others => Comp_Res <= (others => 'X'); Resultado <= (others => 'X');
40         end case;
41     end process;
42 end Behavioral;
```

8.1.1.6. Comparação

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3
4 entity Comparacao is
5     Port ( Comp_Res : in  STD_LOGIC_VECTOR (4 downto 0);
6           Comp_Flag : in  STD_LOGIC;
7           Clk : in  STD_LOGIC;
8           Sel_Comp : in  STD_LOGIC_VECTOR (2 downto 0);
9           S_Flag : out  STD_LOGIC);
10 end Comparacao;
11
12 architecture Behavioral of Comparacao is
13     begin
14         process (Clk, Sel_Comp, Comp_Flag, Comp_Res)
15             variable mem : STD_LOGIC_VECTOR (4 downto 0);
16             begin
17                 case Sel_Comp is
18                     when "000" => S_Flag <= mem(0);
19                     when "001" => S_Flag <= mem(1);
20                     when "010" => S_Flag <= mem(2);
21                     when "011" => S_Flag <= mem(3);
22                     when "100" => S_Flag <= mem(4);
23                     when others => S_Flag <= 'X';
24                 end case;
25                 if rising_edge (Clk) then
26                     if Comp_Flag = '1' then
27                         mem := Comp_Res;
28                     end if;
29                 end if;
30             end process;
31 end Behavioral;
```

8.1.1.7. Contador de programa (PC)

```
1 library IEEE;
2 use IEEE.STD_LOGIC_1164.ALL;
3 use IEEE.STD_LOGIC_UNSIGNED.ALL;
4
5 entity ProgramCounter is
6     Port ( Constante : in  STD_LOGIC_VECTOR (7 downto 0);
7           ESCR_PC,Clock,Reset : in  STD_LOGIC;
8           Endereco : out  STD_LOGIC_VECTOR (7 downto 0));
9 end ProgramCounter;
10
11 architecture Behavioral of ProgramCounter is
12     Signal contagem : STD_LOGIC_VECTOR (7 downto 0);
13     begin
14         process (Clock)
15             begin
16                 if rising_edge (Clock) then
17                     if (Reset = '1') then
18                         contagem <= (others=>'0');
19                     else
20                         if (ESCR_PC = '1') then
21                             contagem<=Constante;
22                         else
23                             contagem<=contagem+"00000001";
24                         end if;
25                     end if;
26                 end if;
27             end Process;
28     Endereco<=contagem;
29 end Behavioral;
```


8.1.1.8. Multiplexer do Program Counter (Mux_PC)

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity Mux_PC is
5      Port ( S_FLAG : in  STD_LOGIC;
6            Operandol : in  STD_LOGIC_VECTOR (7 downto 0);
7            SEL_PC : in  STD_LOGIC_VECTOR (2 downto 0);
8            ESCR_PC : out  STD_LOGIC);
9  end Mux_PC;
10
11 architecture Behavioral of Mux_PC is
12
13 begin
14     process (S_FLAG, Operandol, SEL_PC)
15     begin
16         case SEL_PC is
17             when "000" => ESCR_PC <= '0';
18             when "001" => ESCR_PC <= '1';
19             when "010" => ESCR_PC <= S_FLAG;
20             when "011" => ESCR_PC <= Operandol(7);
21             when "100" => ESCR_PC <= not(Operandol(7) or Operandol(6) or Operandol(5) or
22             Operandol(4) or Operandol(3) or Operandol(2) or Operandol(1) or Operandol(0));
23             when others => ESCR_PC <= 'X';
24         end case;
25     end process;
26
27 end Behavioral;
```

8.1.1.9. ROM de decodificação (ROM)

```
1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity ROM_Descodificacao is
5      Port ( Opcode : in  STD_LOGIC_VECTOR (4 downto 0);
6            Sel_ALU : out  STD_LOGIC_VECTOR (2 downto 0);
7            Escr_Perif_Saida : out  STD_LOGIC;
8            Sel_Data : out  STD_LOGIC_VECTOR (1 downto 0);
9            Escr_Registo : out  STD_LOGIC;
10           WR : out  STD_LOGIC;
11           Sel_PC : out  STD_LOGIC_VECTOR (2 downto 0);
12           Comp_Flag : out  STD_LOGIC;
13           Sel_Comp : out  STD_LOGIC_VECTOR (2 downto 0));
14 end ROM_Descodificacao;
15
16 architecture Behavioral of ROM_Descodificacao is
17
18 begin
19     process (Opcode)
20     begin
21         case Opcode is
22             -- LDP Ri
23             when "00000" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "01";
24             Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
25             -- STP RA
26             when "00001" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '1'; Sel_Data <= "XX";
27             Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
28             -- LD Ri, constante
29             when "00010" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "11";
30             Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
31             -- LD Ri, [constante]
32             when "00011" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "10";
33             Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
34             -- ST [constante], RA
35             when "00100" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
36             Escr_Registo <= '0'; WR <= '1'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
37             -- ADD RA,RB
38             when "00101" => Sel_ALU <= "000"; Escr_Perif_Saida <= '0'; Sel_Data <= "00";
39             Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
40             -- SUB RA,RB
41             when "00110" => Sel_ALU <= "001"; Escr_Perif_Saida <= '0'; Sel_Data <= "00";
42             Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <= '0'; Sel_Comp <= "XXX";
```

```

43      -- AND RA,RB
44      when "00111" => Sel_ALU <= "010"; Escr_Perif_Saida <= '0'; Sel_Data <= "00";
45      Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <='0'; Sel_Comp <= "XXX";
46      -- OR RA,RB
47      when "01000" => Sel_ALU <= "011"; Escr_Perif_Saida <= '0'; Sel_Data <= "00";
48      Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <='0'; Sel_Comp <= "XXX";
49      -- XOR RA,RB
50      when "01001" => Sel_ALU <= "100"; Escr_Perif_Saida <= '0'; Sel_Data <= "00";
51      Escr_Registo <= '1'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <='0'; Sel_Comp <= "XXX";
52      -- CMP RA,RB
53      when "01010" => Sel_ALU <= "101"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
54      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <='1'; Sel_Comp <= "XXX";
55      -- JG constante
56      when "01011" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
57      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "010"; Comp_Flag <='0'; Sel_Comp <= "000";
58      -- JGE constante
59      when "01100" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
60      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "010"; Comp_Flag <='0'; Sel_Comp <= "001";
61      -- JE constante
62      when "01101" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
63      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "010"; Comp_Flag <='0'; Sel_Comp <= "010";
64      -- JLE constante
65      when "01110" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
66      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "010"; Comp_Flag <='0'; Sel_Comp <= "011";
67      -- JL constante
68      when "01111" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
69      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "010"; Comp_Flag <='0'; Sel_Comp <= "100";
70      -- JMP constante
71      when "10000" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
72      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "001"; Comp_Flag <='0'; Sel_Comp <= "XXX";
73      -- JN RA, constante
74      when "10001" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
75      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "011"; Comp_Flag <='0'; Sel_Comp <= "XXX";
76      -- JZ RA, constante
77      when "10010" => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
78      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "100"; Comp_Flag <='0'; Sel_Comp <= "XXX";
79      -- NOP
80      when others => Sel_ALU <= "XXX"; Escr_Perif_Saida <= '0'; Sel_Data <= "XX";
81      Escr_Registo <= '0'; WR <= '0'; Sel_PC <= "000"; Comp_Flag <='0'; Sel_Comp <= "XXX";
82      end case;
83      end process;
84      end Behavioral;

```

8.1.2. Memória de Instruções

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3
4  entity Memoria_Instrucoes is
5      Port ( Endereco : in  STD_LOGIC_VECTOR (7 downto 0);
6            Opcode : out  STD_LOGIC_VECTOR (4 downto 0);
7            Sel_R : out  STD_LOGIC;
8            Constante : out  STD_LOGIC_VECTOR (7 downto 0));
9  end Memoria_Instrucoes;
10
11  architecture Behavioral of Memoria_Instrucoes is
12
13  begin
14      process (Endereco)
15      begin
16          case Endereco is
17              when "00000000" => Opcode <= "00010"; Sel_R <= '0'; Constante <= "000000011";
18              when "00000001" => Opcode <= "00100"; Sel_R <= '0'; Constante <= "000000000";
19              when "00000010" => Opcode <= "00010"; Sel_R <= '0'; Constante <= "000101000";
20              when "00000011" => Opcode <= "00100"; Sel_R <= '0'; Constante <= "000000001";
21              when "00000100" => Opcode <= "00000"; Sel_R <= '0'; Constante <= "XXXXXXXXX";
22
23              when "00000101" => Opcode <= "10001"; Sel_R <= 'X'; Constante <= "000101100";
24              when "00000110" => Opcode <= "00010"; Sel_R <= '1'; Constante <= "001010000";
25              when "00000111" => Opcode <= "01010"; Sel_R <= 'X'; Constante <= "XXXXXXXXX";
26              when "00001000" => Opcode <= "01100"; Sel_R <= 'X'; Constante <= "000110111";
27              when "00001001" => Opcode <= "00100"; Sel_R <= '0'; Constante <= "000000100";
28
29              when "00001010" => Opcode <= "00100"; Sel_R <= '0'; Constante <= "000000011";
30              when "00001011" => Opcode <= "00011"; Sel_R <= '0'; Constante <= "000000000";
31              when "00001100" => Opcode <= "00010"; Sel_R <= '1'; Constante <= "000000001";
32              when "00001101" => Opcode <= "00110"; Sel_R <= '0'; Constante <= "XXXXXXXXX";
33              when "00001110" => Opcode <= "10010"; Sel_R <= 'X'; Constante <= "000101000";
34
35              when "00001111" => Opcode <= "00100"; Sel_R <= '0'; Constante <= "000000000";
36              when "00010000" => Opcode <= "00011"; Sel_R <= '0'; Constante <= "000000011";
37              when "00010001" => Opcode <= "00011"; Sel_R <= '1'; Constante <= "000000100";
38              when "00010010" => Opcode <= "00101"; Sel_R <= '0'; Constante <= "XXXXXXXXX";
39              when "00010011" => Opcode <= "10000"; Sel_R <= 'X'; Constante <= "000010100";
40

```

```

41         when "00010100" => Opcode <= "00011"; Sel_R <= '0'; Constante <= "000000011";
42         when "00010101" => Opcode <= "10000"; Sel_R <= 'X'; Constante <= "00011101";
43         when "00010110" => Opcode <= "00010"; Sel_R <= '1'; Constante <= "11111111";
44         when "00010111" => Opcode <= "01001"; Sel_R <= '0'; Constante <= "XXXXXXXX";
45         when "00011000" => Opcode <= "00010"; Sel_R <= '1'; Constante <= "00000001";
46
47         when "00011001" => Opcode <= "00101"; Sel_R <= '0'; Constante <= "XXXXXXXX";
48         when "00011010" => Opcode <= "10000"; Sel_R <= 'X'; Constante <= "00011101";
49         when "00011011" => Opcode <= "00011"; Sel_R <= '1'; Constante <= "00000001";
50         when "00011100" => Opcode <= "00110"; Sel_R <= '0'; Constante <= "XXXXXXXX";
51         when "00011101" => Opcode <= "00001"; Sel_R <= '0'; Constante <= "XXXXXXXX";
52         when "00011110" => Opcode <= "10000"; Sel_R <= 'X'; Constante <= "00011110";
53
54         when others => Opcode <= "XXXXX"; Sel_R <= 'X'; Constante <= "XXXXXXXX";
55     end case;
56 end process;
57 end Behavioral;

```

8.1.3. Memória de Dados (RAM)

```

1  library IEEE;
2  use IEEE.STD_LOGIC_1164.ALL;
3  use IEEE.NUMERIC_STD.ALL;
4
5  entity RAM is
6      Port ( Operand1, Address : in  STD_LOGIC_VECTOR (7 downto 0);
7            WR, Clock : in  STD_LOGIC;
8            Dados_M : out  STD_LOGIC_VECTOR (7 downto 0));
9  end RAM;
10
11  architecture Behavioral of RAM is
12      Type mem is array(256 downto 0) of STD_LOGIC_VECTOR(7 Downto 0);
13      Signal Memoria : mem := (others=>(others=>'0'));
14
15      begin
16          process(Clock)
17          begin
18              if rising_edge(Clock) then
19                  if WR = '1' then
20                      Memoria(to_integer(unsigned(Address)))<=Operand1;
21                  end if;
22              end if;
23          end process;
24
25      Dados_M<=Memoria(to_integer(unsigned(Address)));
26
27  end Behavioral;

```